

Monte Carlo Simulation in the Pricing of Derivatives

Cara M. Marshall

Fordham University

Queens College of the City University of New York

Monte Carlo Simulation in the Pricing of Derivatives

Of all financial products, the most difficult to price are the derivatives. Of course, some derivatives are far more difficult to price than others. The models used to price derivatives can be divided into three broad categories: analytical models, numeric models, and simulation models.¹ The latter refers to Monte Carlo simulation, named after a famous casino in Monaco on the French Riviera. Monte Carlo simulation was first used to estimate the probability of winning a game of pure chance. The technique works by replicating the outcomes of a stochastic process through the careful use of random numbers. As the number of replications increases, the resulting range of approximation, represented by the average, narrows, converging to the analytically correct solution. Since Monte Carlo simulation generally requires a significant number of repetitions of an algorithm (a massive number of calculations), its practical application requires a computer with a fast CPU. Monte Carlo simulation can be used in all sorts of business applications whenever there is a source of uncertainty (such as future stock prices, interest rates, exchange rates, commodity prices, etc.). To illustrate the basic concepts, we will focus on pricing options, which are generally the most difficult types of derivatives to value.

The classic example of an analytical model used to price options is the Black-Scholes model published by Fischer Black and Myron Scholes in 1973. The classic example of a numeric model to price options is the binomial option pricing model published by John Cox, Stephen Ross, and Mark Rubinstein in 1979.

¹ Many people consider simulation models to be a subset of numeric models. I believe that they are sufficiently different to be in a class by themselves, which is typically the way they are viewed by the financial services firms that use simulation models so effectively.

Analytical models are the most elegant of the pricing methodologies. In this approach, we begin with a set of assumptions about how the relevant variables behave. We then translate these assumptions into mathematical equations. We then use the mathematical equations to derive a formal relationship between the input variables and the output variable (in this case the output variable is the fair option premium). The analytical model is the end result of the derivation process and takes the form of a “formula” or “equation” that ties the inputs to the output. This formula is often referred to as the “solution.” The beauty of analytical models is that they allow us to quickly produce a precise valuation. But there are problems associated with analytical models as well. First, the models are quite difficult to derive without an advanced knowledge of stochastic calculus. Second, in some cases it is not possible to derive analytical solutions. Third, analytical models once derived are completely inflexible. That is, they can only be applied in situations where the exact set of assumptions used to build the model hold.

Numeric models are much more flexible than analytical models. In these models one again lays out a set of assumptions, but rather than deriving an equation that ties the inputs to the output, the model employs a finite series of steps—in an algorithmic sense—to arrive at a value. These models provide approximations of the true value rather than a precise solution. The benefit of numeric models is that they are relatively simple to build and do not require a deep understanding of stochastic calculus to appreciate them or to interpret them. They are also much more flexible than analytical models. For example, one can often easily change the assumptions of the model to fit new situations. The downside is that the degree of precision is directly related to the number of computations (steps) one is prepared to employ. To get a very high degree of

precision, the user might have to perform tens of millions of calculations. This drawback of numeric models however has been rendered largely moot by the tremendous increase in the speed of microprocessors over the years and by the realization that there are often shortcuts that can be employed with no loss of precision.

Simulation models are not as elegant as analytical models, and not as fast as either analytical models or numeric models. Their great strength is that they are incredibly flexible and can be used to value derivatives that do not easily lend themselves to analytical valuation techniques or numeric valuation techniques. Indeed, the future price of any financial asset can be simulated when it is expressed in the form of an expected value.² In the simulation approach, we program a computer to “simulate” observations on the random variable of interest. Care must be taken to ensure that the simulated values of the variable possess the distributional properties and statistical parameters that we require.

The easiest way to see how this is done is to walk through a few exercises. We will walk through three. In the first exercise, we will build a Monte Carlo simulator to value a plain vanilla call and a plain vanilla put under the same assumptions that Black and Scholes employed. Such options are written on the “terminal price of the underlying.” This will be a good test of how accurate such a model can be. In our second exercise, we will value a call and a put written on a stock’s “price return” measured over the life of the option, rather than on the stock’s terminal price. The former typically trade on option exchanges and the latter typically trade in the OTC options market. Finally, in our third exercise, we will build a simulator to value an option written

² Elaboration of this point can be found in Briys, et al (1998).

on more than one underlying asset. It is in this last case that we begin to really see the strength of simulation. As we go along, each exercise will build on the one before it.

Pricing a Classic Black-Scholes Option:

Black and Scholes assumed (1) that the future price of the underlying stock is distributed lognormally; (2) that the risk-free rate of interest is constant and the same for all maturities; (3) that the stock's volatility is constant; (4) that the option is only exercisable at the end of the option's life (i.e., it is European type); and (5) that the underlying stock does not pay any dividends. There are a few other assumptions as well, but they are not necessary for this exercise.

Let's suppose that the current spot price of the underlying stock is \$100, that the option is struck at the money (strike price = \$100), that the option has 3 months to expiration (.25 years), that the annual interest rate is 5%, and that the underlying stock's volatility is 30%. For purposes of later comparison, the exact Black-Scholes value for the call is \$6.5831 and the exact Black-Scholes value for the put is \$5.3508.

In a simulation approach, we need to "simulate" observations on the terminal price of the stock. By terminal price of the stock, we mean the possible values of the stock at the point in time when the option is due to expire. These observations must have the right type of distribution (lognormal), the right mean, and the right standard deviation. Once we have simulated a possible terminal value for the stock's price at the point of the option's expiration (S_E) we can then determine the terminal value of the option. If the option is a call, its terminal value is given by $\max[S_E - X, 0]$ where X denotes the strike price. If the option is a put, the terminal value of the option is given by $\max[X - S_E, 0]$. Once we have the terminal value of the option, we then discount it back to the present at the risk-free rate of interest to obtain the present value that corresponds to the terminal

value. This present value is just one of an infinite number of possible present values the option might have because there are an infinite number of terminal values that the option might have. We then repeat this exercise many times. It is not unusual to run a simulator 100,000 times or more. Suppose that we run the simulator just 50,000 times. While each present value we generate is only one possible result, the average of the runs of the simulator should come very close to the true value of the option.

Let's do this simulation using the ubiquitous Microsoft product Excel[®]. At each step in the process we will show the relevant formula that goes into the spreadsheet cell so that you can replicate the model as we go along. There will be a "box" around the cell whose formula appears in the formula field.

Excel[®] comes with one built in random number generator. This is the function RAND(). The RAND() function generates random numbers having a uniform continuous distribution bounded between 0 and 1 (with a mean of 0.5). This is not the type of distribution we want, but we can get to the type of distribution we want through a series of steps starting with the RAND() function. Excel[®] also has a function that is designed to generate an observation drawn from a standard normal distribution if you enter a cumulative probability. This is the NORMSINV() function. Since a cumulative probability must be between 0 and 1, we can generate observations on a standard normal distribution by embedding the RAND() function in the NORMSINV() function: NORMSINV(RAND()). Call this value Z.

We will illustrate the building process step by step including "screen shots." Because we are employing a random number generator, at each step, the outputs that are

driven by the random number generator will change automatically. This might give the impression that there is inconsistency between the different screen shots, but there is not.

B17		fx =NORMSINV(RAND())					
	A	B	C	D	E	F	G
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts						
2							
3							
4	Stock Price=	100.00					
5	Interest Rate =	5.000%					
6	Volatility (annual)=	30.00%					
7	Call Strike=	100.00					
8	Put Strike=	100.00					
9	Time to Expiry (in years)=	0.2500					
14							
15		N(0,1)					
16	Observation	Z					
17	1	1.0573					
18							

Now that we have randomly generated observations from a standard normal distribution (with a mean of 0 and a standard deviation of 1), we need to convert them to observations from a non-standard normal distribution. Suppose that $Y \sim N(\mu, \sigma)$ and define Z as $\frac{Y - \mu}{\sigma}$ so that $Z \sim N(0,1)$. This implies that $Y = Z \times \sigma + \mu$.

The non-standard normal distribution would have a periodic mean (i.e., mean for the relevant period) equal to the expected growth rate in the stock's price over the life of the option. This is known as the drift factor or drift rate.³ Black and Scholes discovered that, in the context of options, a stock's expected growth rate is independent of its expected return and dependent only on the risk-free rate of interest and the stock's

³ The drift factor is the average increase per unit of time in a stochastic variable. Derivation of the drift factor formula is beyond the scope of this demonstration. For those who would like to pursue it more deeply, see John Hull *Options, Futures & Other Derivatives*, 6th edition 2005, Wiley, Chapter 12.

volatility. This is counterintuitive, and is one of the key features of the Black-Scholes result, commonly known as the risk-neutrality assumption. It is important that our simulation model be consistent with the Black-Scholes framework. The drift factor (or mean) is given by:

$$\mu_{per} = (r - \frac{1}{2}\sigma^2)\tau$$

Where r denotes the risk-free rate compounded continuously, σ is the annual volatility and τ is the time to option expiry, measured in years. For a three month option, $\tau = 0.25$.

B11		fx = (\$B\$5-(0.5*(\$B\$6^2)))*\$B\$9					
	A	B	C	D	E	F	G
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts						
2							
3							
4	Stock Price=	100.00					
5	Interest Rate =	5.000%					
6	Volatility (annual)=	30.00%					
7	Call Strike=	100.00					
8	Put Strike=	100.00					
9	Time to Expiry (in years)=	0.2500					
10							
11	periodic mean (drift factor)	0.125%					

The standard deviation is simply the periodic volatility for the underlying stock measured over the life of the option and is simply the square root of the time to option expiry multiplied by the annual volatility:

$$\sigma_{per} = \sigma\sqrt{\tau}$$

B12		fx = \$B\$6*(\$B\$9^0.5)					
	A	B	C	D	E	F	G
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts						
2							
3							
4	Stock Price=	100.00					
5	Interest Rate =	5.000%					
6	Volatility (annual)=	30.00%					
7	Call Strike=	100.00					
8	Put Strike=	100.00					
9	Time to Expiry (in years)=	0.2500					
10							
11	periodic mean (drift factor)	0.125%					
12	periodic standard deviation	15.000%					

Plugging these values for the mean and standard deviation into $Y = Z \times \sigma + \mu$ would give us one possible *percentage change* in the price of the underlying stock measured on the assumption of continuous compounding. Call this value Y .

C17		fx = (B17*\$B\$12)+\$B\$11					
	A	B	C	D	E	F	G
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts						
2							
3							
4	Stock Price=	100.00					
5	Interest Rate =	5.000%					
6	Volatility (annual)=	30.00%					
7	Call Strike=	100.00					
8	Put Strike=	100.00					
9	Time to Expiry (in years)=	0.2500					
10							
11	periodic mean (drift factor)	0.125%					
12	periodic standard deviation	15.000%					
13							
14							
15		N(0,1)	N(m,s)				
16	Observation	Z	Y				
17	1	2.7713	0.4169				
18							

The next step is to convert this percentage change in the stock price into a terminal price for the stock. Keeping in mind that the percentage change in the price of

the stock is measured on the assumption of continuous compounding, the conversion to a terminal stock price is: $S_E = \exp(Y) \times S_0$.

Besides giving us the terminal price, this conversion also guarantees that the resultant terminal price will be lognormally distributed. The reason for this is straightforward. If a random variable has a lognormal distribution, then the natural log of that random variable has a normal distribution. Similarly, if a random variable has a normal distribution, the exponential of that random variable will have a lognormal distribution (since the exponential function is the inverse of the natural logarithm function).

D17		fx =EXP(C17)*\$B\$4					
	A	B	C	D	E	F	G
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts						
2							
3							
4	Stock Price=	100.00					
5	Interest Rate =	5.000%					
6	Volatility (annual)=	30.00%					
7	Call Strike=	100.00					
8	Put Strike=	100.00					
9	Time to Expiry (in years)=	0.2500					
10							
11	periodic mean (drift factor)	0.125%					
12	periodic standard deviation	15.000%					
13							
14							
15		N(0,1)	N(m,s)				
16	Observation	Z	Y	S			
17	1	0.9552	0.1445	115.5489			
18							

Once we have a simulated terminal price for the stock, we can compute the simulated terminal values for the options using the MAX() function to take the maximum value.

$$\text{Call}_E = \max[S_E - X, 0]$$

$$\text{Put}_E = \max[X - S_E, 0].$$

E17 fx =MAX(D17-\$B\$7,0)									
	A	B	C	D	E	F	G	H	I
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts								
2									
3									
4	Stock Price=	100.00							
5	Interest Rate =	5.000%							
6	Volatility (annual)=	30.00%							
7	Call Strike=	100.00							
8	Put Strike=	100.00							
9	Time to Expiry (in years)=	0.2500							
10									
11	periodic mean (drift factor)	0.125%							
12	periodic standard deviation	15.000%							
13									
14									
15									
16	Observation	N(0,1) Z	N(m,s) Y	S	Call Terminal Value			Put Terminal Value	
17	1	0.3870	0.0593	106.1090	6.1090			0.0000	
18									

Once we have calculated the terminal values, we can simply discount them back to their present value. Under continuous compounding, the discounting equation is:

$$\text{PV} = \exp(-\tau \times r) \times \text{Terminal Value}.$$

F17 fx =EXP(-(\$B\$5*\$B\$9))*E17									
	A	B	C	D	E	F	G	H	I
1	Monte Carlo Simulation Analysis: Valuing Plain Vanilla Calls and Puts								
2									
3									
4	Stock Price=	100.00							
5	Interest Rate =	5.000%							
6	Volatility (annual)=	30.00%							
7	Call Strike=	100.00							
8	Put Strike=	100.00							
9	Time to Expiry (in years)=	0.2500							
10									
11	periodic mean (drift factor)	0.125%							
12	periodic standard deviation	15.000%							
13									
14									
15									
16	Observation	N(0,1) Z	N(m,s) Y	S	Call Terminal Value	Present Value		Put Terminal Value	Present Value
17	1	0.1388	0.0221	102.2314	2.2314	2.2037		0.0000	0.0000
18									

Now that we have simulated one possible present value for the call and one possible present value for the put, we will run the simulator many times and take an average. The more times we run the simulation to obtain the average, the closer we should get to the true value of the option (essentially an application of the law of large numbers).

Simulation Results

Observations	Call Price	Deviation from Black-Scholes	Put Price	Deviation from Black-Scholes
100	\$7.40	\$0.82	\$4.36	\$-0.98
1,000	\$6.92	\$0.34	\$5.13	\$-0.21
5,000	\$6.47	\$-0.11	\$5.46	\$0.12
25,000	\$6.63	\$0.05	\$5.36	\$0.02
50,000	\$6.55	\$-.03	\$5.36	\$0.02

In order to run the simulation 50,000, we would simply copy the formula 50,000 times and take a simple average of the simulated values. Depending on the speed of your processor, it may take a few moments to run the simulation due to the large number of calculations involved. As a side note, each worksheet in an Excel[®] file (workbook) contains 65,536 rows, so our number of simulations is limited. But, if you wanted to run more, say 100,000 simulations, you could run the 50,000 simulations twice and take an average of the two averages to replicate the results of 100,000 simulations. It would be more efficient, however, to write macros that automate the process. We will stick with 50,000 for purposes of this demonstration.

Price Return:

Equity options that trade on exchanges are written on the stock's price. That is, the payoff at the end is dependent on the stock's price and the strike price of the option. Up until now, we have talked about equity options as though they are written on one share of stock. Of course, options are typically written on more than one share. Exchange traded options are usually written on 100 shares. Let's denote the number of shares by Q . Then the actual payoff on the options we have been discussing would be given by:

$$\text{Payoff}_{\text{call}} = \max[S_E - X, 0] \times Q$$

$$\text{Payoff}_{\text{put}} = \max[X - S_E, 0] \times Q$$

Unlike exchange traded options, equity options that trade over-the-counter are often written on price return, rather than price. That is, the payoff at the end of the life of the option is given by $\max[\text{PR} - \text{XR}, 0] \times \text{NP}$ for calls and $\max[\text{XR} - \text{PR}, 0] \times \text{NP}$ for puts. Here, PR is the price return, defined as the percentage change in the price of the stock (or stock index) over the life of the option, XR is the "strike rate," and NP is the notional principal. In the case of an at-the-money call, this would be $\max[\text{PR} - 0\%, 0] \times \text{NP}$.

To see that option payoffs defined in terms of terminal stock price and those defined in terms of price return are really the same thing, we will demonstrate that we can easily move from one to the other:

$$\text{Payoff}_{\text{call}} = \max[S_E - X, 0] \times Q$$

If we multiply by $\frac{S_0}{S_0}$:

$$= \max[S_E - X, 0] \times Q \times \frac{S_0}{S_0}$$

$$S_E = \frac{\max[S_E - X, 0]}{S_0} \times Q \times S_0$$

$$= \max\left[1 + PR - \left(1 + \frac{X}{S_0}\right), \frac{0}{S_0}\right] \times NP \quad \text{where } PR = \frac{S_E}{S_0} - 1$$

$$= \max[PR - XR, 0] \times NP \quad \text{where } XR = \frac{X}{S_0} - 1$$

We can easily adapt the simulation model we previously built to price this option. In this exercise, the price of the option will be quoted as a percentage of the notional principal on which the option is written. That is, we will work in terms of \$1 of notional principal. Instead of inputting strike prices, we would input the strike rates for the call and the put. The price return can then be calculated by taking the terminal stock price and dividing it by the beginning stock price. The terminal values for the call and put are then calculated by $\max[PR - XR, 0]$ and $\max[XR - PR, 0]$, respectively. The final step is to discount to obtain the present value. This is done the same way as for the earlier model. In this particular case, with 50,000 simulations in the run, we obtained a price for the call of 6.542% and a price for the put of 5.786%. In both cases, these are interpreted as percentages of the notional principal on which the options are written. For example, if the client wanted to buy the call on \$1,000,000 of notional principal, the price of the option would be \$65,420.

E19 =D19/\$B\$4-1											
	A	B	C	D	E	F	G	H	I	J	K
1	Monte Carlo Simulation Analysis: Valuing Over-The-Counter Calls and Puts (Price Return)										
2											
3											
4	Stock Price=	100.00									
5	Interest Rate =	5.000%									
6	Volatility (annual)=	30.00%									
7	Call Strike Rate=	0.00%									
8	Put Strike Rate=	0.00%									
9	Time to Expiry (in years)=	0.2500									
10											
11	periodic mean (drift factor)	0.125%									
12	periodic standard deviation	15.000%									
13	Simulated Call Value=	6.542%									
14	Simulated Put Value=	5.786%									
15											
16											
17											
18	Observation	N(0,1) Z	N(m,s) Y	S	Price Return		Call Terminal Value	Call Present Value		Put Terminal Value	Put Present Value
19	1	-1.7797	-0.2657	76.6664	-23.33%		0.00%	0.00%		23.33%	23.04%
20	2	0.9604	0.1453	115.6390	15.64%		15.64%	15.44%		0.00%	0.00%
21	3	2.0371	0.3068	135.9089	35.91%		35.91%	35.46%		0.00%	0.00%
22	4	0.6349	0.0965	110.1291	10.13%		10.13%	10.00%		0.00%	0.00%
23	5	0.6456	0.0981	110.3069	10.31%		10.31%	10.18%		0.00%	0.00%
24	6	-1.6119	-0.2405	78.6210	-21.38%		0.00%	0.00%		21.38%	21.11%
25	7	0.2071	0.0323	103.2843	3.28%		3.28%	3.24%		0.00%	0.00%
26	8	-0.3437	-0.0503	95.0942	-4.91%		0.00%	0.00%		4.91%	4.84%
27	9	-0.8433	-0.1252	88.2284	-11.77%		0.00%	0.00%		11.77%	11.63%
28	10	1.0942	0.1654	117.9841	17.98%		17.98%	17.76%		0.00%	0.00%

You will have noticed that, with the proper interpretation, options written on an asset's price and options written on an asset's price return, are really the same thing. However, a key advantage of modeling an option on price returns is that it more easily lends itself to pricing options written on multiple underlyings. We take these up next in the context of rainbow options. The term "rainbow option" is an industry term for options with more than one underlying asset. If there are two underlying assets, they are often called two-color rainbows. If there are three underlying assets, they are called three-color rainbows, and so forth. These should not be confused with index options, that have only one underlying – the index. Rainbows are typically written on price returns and fall within the realm of "exotic options." It is in the valuation of exotic options that simulation shows its real strength.

Pricing Rainbow Option:

Rainbow options come in a number of different varieties and each has its own purposes. We will illustrate them with two underlyings (but there can be any number of underlyings). Consider just a few of the possible payoff structures. Rather than give them formal names, we will simply refer to them as Types 1 through 6.

Type 1:	Payoff = $\max[PR_1, PR_2]$
Type 2:	Payoff = $\max[PR_1, PR_2, 0]$
Type 3:	Payoff = $\max[\max(PR_1, PR_2) - XR, 0]$
Type 4:	Payoff = $\max[\max(PR_1 - XR_1, PR_2 - XR_2), 0]$
Type 5:	Payoff = $\max[\min(PR_1, PR_2) - XR, 0]$
Type 6:	Payoff = $\max[\min(PR_1 - XR_1, PR_2 - XR_2), 0]$

Type 1, often called a “best of” or “better of” option, pays off based on the best performing of two underlyings. But, if both underlyings produce a negative price return, the payoff is actually negative (not zero as in a conventional option). According to Smithson (1998), a popular combination in the early 1990s was a Type 1 two-color rainbow option based on the performance of a stock market index and a bond market index. An investor struggling with the decision of whether to buy stocks or bonds, could buy a Type 1 option on which the payoff is determined by which of the two asset classes performs better.

Type 2 is similar to Type 1, except that it guarantees that the payoff will never be negative. Notice that Type 1 and Type 2 do not provide for a strike rate. Type 3 does allow for a strike rate but the same strike rate applies to both underlyings. If the strike rate is set to zero, then Type 3 collapses to Type 2. Type 4 allows for more than one strike rate, i.e., a different strike rate can be applied to each of the underlyings (often called a dual-strike rainbow). However, if the two strikes happen to be the same, Type 4 collapses to Type 3. Types 5 and 6 are analogous to Types 3 and 4 except that Types 5 and 6 pay off based on the worst performer, rather than the best performer. All six of these options can be thought of as calls. We will build a simulation model to price up Types 4 and 6.

Suppose that we are going to write a single option on two U.S. stocks. Let's call these stocks ABC and XYZ. In one case, the payoff would be based on the better performer (Type 4) and in the other case, the payoff would be based on the worst performer (Type 6). For simplicity, assume that the option is written at-the-money with respect to both of the underlying stocks and neither stock pays dividends. The payoff, per \$1 of notional principal, for a call of Type 4 at its expiry would be

$\max[\max(\text{PR}_{\text{ABC}} - 0\%, \text{PR}_{\text{XYZ}} - 0\%), 0]$. And the payoff for a call of Type 6 at its expiry would be $\max[\min(\text{PR}_{\text{ABC}} - 0\%, \text{PR}_{\text{XYZ}} - 0\%), 0]$.

What complicates things now, relative to our previous example of an option written on a single price return, is that we have an additional value driver. This additional value driver is the correlation of the price returns on the two stocks. How the correlation will impact the value of a rainbow option depends on the nature of the payoff function. This correlation has to be included in the simulated terminal stock values.

To have a concrete example to work with, let's suppose that ABC is currently priced at \$100 a share and has an annual volatility of 25%, and XYZ is currently priced at \$50 a share and has an annual volatility of 45%. The options both have three months to expiry ($\tau = 0.25$ years). As already noted, for simplicity, we will assume that the options are written at-the-money with respect to both stocks so that the strike rates XR_1 and XR_2 are both zero. The annual interest rate is 5% and the degree of correlation between the two stocks' returns is 0.65.

We begin as we did in the last model, except that instead of generating observations on one standard normal random variable, we have to simultaneously generate observations on two separate standard normal random variables. In both cases we use the `NORMSINV(RAND())` function. Denote these Z_1 and Z_2 . These two variables are uncorrelated with each other. They must therefore be "adjusted" to bring in the correlation. This is a simple procedure that employs a well known statistical relationship. Define two new random variables R_1 and R_2 and let $R_1 = Z_1$ and $R_2 = \rho Z_1 + Z_2 \sqrt{1 - \rho^2}$ where ρ denotes the degree of correlation between the two stocks' returns. R_1 and R_2 are now standard normal random variables (with means of 0 and standard deviations of 1) and they have the desired degree of correlation.

E22		fx =(\$B\$10*B22)+(C22*SQRT((1-(\$B\$10^2))))				Q	R	S
	A	B	C	D	E			
1	Monte Carlo Simulation Analysis: Valuing a Rainbow Option of Type 4 and Type 6							
2								
3		Stock 1	Stock 2					
4	Stock Price=	100.00	50.00					
5	Interest Rate =	5.000%						
6	Volatility (annual)=	25.00%	45.00%					
7	Type 4 Strike Rate=	0.00%	0.00%					
8	Type 6 Strike Rate=	0.00%	0.00%					
9	Time to Expiry (in years)=	0.2500						
10	Correlation=	0.6500						
11								
12	periodic mean (drift factor)	0.469%	-1.281%					
13	periodic standard deviation	12.500%	22.500%					
14								
19								
20		N(0,1)	N(0,1)	N(0,1)	N(0,1)			
21	Observation	Z1	Z2	R1	R2			
22	1	-1.5312	0.0223	-1.5312	-0.9784			
23	2	-1.3815	1.6232	-1.3815	0.3356			
24	3	1.1933	-0.8405	1.1933	0.1369			
25	4	0.2509	-0.3084	0.2509	-0.0713			
26	5	-0.4047	-0.3212	-0.4047	-0.5072			
50017	49996	-0.5600	-0.6834	-0.5600	-0.8833			
50018	49997	0.4625	-0.4862	0.4625	-0.0689			
50019	49998	-1.1020	-0.3230	-1.1020	-0.9617			
50020	49999	2.5944	0.4642	2.5944	2.0391			
50021	50000	-0.7516	-0.5744	-0.7516	-0.9250			

We now have to convert these standard normal distributions to appropriate non-standard normal distributions. We will denote these as Y_1 and Y_2 . Y_1 and Y_2 will be calculated from R_1 and R_2 by incorporating an appropriate mean and standard deviation for each. The mean, standard deviation, terminal stock price, and price return for each stock are calculated in the same manner as we did in the last exercise. As before, we will run the simulator 50,000 times.

K22		=I22/\$C\$4-1									
	A	B	C	D	E	F	G	H	I	J	K
1	Monte Carlo Simulation Analysis: Valuing a Rainbow Option of Type 4 and Type 6										
2											
3		Stock 1	Stock 2								
4	Stock Price=	100.00	50.00								
5	Interest Rate =	5.000%									
6	Volatility (annual)=	25.00%	45.00%								
7	Type 4 Strike Rate=	0.00%	0.00%								
8	Type 6 Strike Rate=	0.00%	0.00%								
9	Time to Expiry (in years)=	0.2500									
10	Correlation=	0.6500									
11											
12	periodic mean (drift factor)	0.469%	-1.281%								
13	periodic standard deviation	12.500%	22.500%								
14											
19											
20		N(0,1)	N(0,1)	N(0,1)	N(0,1)	N(m,s)	N(m,s)			Price	Price
21	Observation	Z1	Z2	R1	R2	Y1	Y2	S1	S2	Return1	Return2
22	1	-1.5312	0.0223	-1.5312	-0.9784	-0.1867	-0.2329	82.9678	39.6099	-17.03%	-20.78%
23	2	-1.3815	1.6232	-1.3815	0.3356	-0.1680	0.0627	84.5359	53.2354	-15.46%	6.47%
24	3	1.1933	-0.8405	1.1933	0.1369	0.1538	0.0180	116.6310	50.9079	16.63%	1.82%
25	4	0.2509	-0.3084	0.2509	-0.0713	0.0360	-0.0289	103.6703	48.5780	3.67%	-2.84%
26	5	-0.4047	-0.3212	-0.4047	-0.5072	-0.0459	-0.1269	95.5139	44.0400	-4.49%	-11.92%
50017	49996	-0.5600	-0.6834	-0.5600	-0.8833	-0.0653	-0.2116	93.6780	40.4660	-6.32%	-19.07%
50018	49997	0.4625	-0.4862	0.4625	-0.0689	0.0625	-0.0283	106.4498	48.6046	6.45%	-2.79%
50019	49998	-1.1020	-0.3230	-1.1020	-0.9617	-0.1331	-0.2292	87.5413	39.7586	-12.46%	-20.48%
50020	49999	2.5944	0.4642	2.5944	2.0391	0.3290	0.4460	138.9558	78.1017	38.96%	56.20%
50021	50000	-0.7516	-0.5744	-0.7516	-0.9250	-0.0893	-0.2209	91.4609	40.0882	-8.54%	-19.82%

The terminal value equation for the Type 4 call is:

$\max[\max(PR_1 - XR_1, PR_2 - XR_2), 0]$. The terminal value equation for the Type 6 call is

$\max[\min(PR_1 - XR_1, PR_2 - XR_2), 0]$.


L22													=MAX(MAX(J22-\$B\$7,K22-\$C\$7),0)	
	A	B	C	D	E	F	G	H	I	J	K	L	M	
1	Monte Carlo Simulation Analysis: Valuing a Rainbow Option of Type 4 and Type 6													
2														
3		Stock 1	Stock 2											
4	Stock Price=	100.00	50.00											
5	Interest Rate =	5.000%												
6	Volatility (annual)=	25.00%	45.00%											
7	Type 4 Strike Rate=	0.00%	0.00%											
8	Type 6 Strike Rate=	0.00%	0.00%											
9	Time to Expiry (in years)=	0.2500												
10	Correlation=	0.6500												
11														
12	periodic mean (drift factor)	0.469%	-1.281%											
13	periodic standard deviation	12.500%	22.500%											
14														
15	Simulated Type 4 Value=	11.266%												
16	Simulated Type 6 Value=	3.701%												
17														
18														
19														
20														
21	Observation	N(0,1) Z1	N(0,1) Z2	N(0,1) R1	N(0,1) R2	N(m,s) Y1	N(m,s) Y2	S1	S2	Price Return1	Price Return2	Type 4 Terminal Value	Type 4 Present Value	
22	1	-1.5312	0.0223	-1.5312	-0.9784	-0.1867	-0.2329	82.9678	39.6099	-17.03%	-20.78%	0.00%	0.00%	
23	2	-1.3815	1.6232	-1.3815	0.3356	-0.1680	0.0627	84.5359	53.2354	-15.46%	6.47%	6.47%	6.39%	
24	3	1.1933	-0.8405	1.1933	0.1369	0.1538	0.0180	116.6310	50.9079	16.63%	1.82%	16.63%	16.42%	
25	4	0.2509	-0.3084	0.2509	-0.0713	0.0360	-0.0289	103.6703	48.5780	3.67%	-2.84%	3.67%	3.62%	
26	5	-0.4047	-0.3212	-0.4047	-0.5072	-0.0459	-0.1269	95.5139	44.0400	-4.49%	-11.92%	0.00%	0.00%	
50017	49996	-0.5600	-0.6834	-0.5600	-0.8833	-0.0653	-0.2116	93.6780	40.4660	-6.32%	-19.07%	0.00%	0.00%	
50018	49997	0.4625	-0.4862	0.4625	-0.0689	0.0625	-0.0283	106.4498	48.6046	6.45%	-2.79%	6.45%	6.37%	
50019	49998	-1.1020	-0.3230	-1.1020	-0.9617	-0.1331	-0.2292	87.5413	39.7586	-12.46%	-20.48%	0.00%	0.00%	
50020	49999	2.5944	0.4642	2.5944	2.0391	0.3290	0.4460	138.9558	78.1017	38.96%	56.20%	56.20%	55.51%	
50021	50000	-0.7516	-0.5744	-0.7516	-0.9250	-0.0893	-0.2209	91.4609	40.0882	-8.54%	-19.82%	0.00%	0.00%	
50022												Average	11.27%	

O22													=MAX(MIN(J22-\$B\$8,K22-\$C\$8),0)			
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	Monte Carlo Simulation Analysis: Valuing a Rainbow Option of Type 4 and Type 6															
2																
3		Stock 1	Stock 2													
4	Stock Price=	100.00	50.00													
5	Interest Rate =	5.000%														
6	Volatility (annual)=	25.00%	45.00%													
7	Type 4 Strike Rate=	0.00%	0.00%													
8	Type 6 Strike Rate=	0.00%	0.00%													
9	Time to Expiry (in years)=	0.2500														
10	Correlation=	0.6500														
11																
12	periodic mean (drift factor)	0.469%	-1.281%													
13	periodic standard deviation	12.500%	22.500%													
14																
15	Simulated Type 4 Value=	11.266%														
16	Simulated Type 6 Value=	3.701%														
17																
18																
19																
20																
21	Observation	N(0,1) Z1	N(0,1) Z2	N(0,1) R1	N(0,1) R2	N(m,s) Y1	N(m,s) Y2	S1	S2	Price Return1	Price Return2	Type 4 Terminal Value	Type 4 Present Value	Type 6 Terminal Value	Type 6 Present Value	
22	1	-1.5312	0.0223	-1.5312	-0.9784	-0.1867	-0.2329	82.9678	39.6099	-17.03%	-20.78%	0.00%	0.00%	0.00%	0.00%	
23	2	-1.3815	1.6232	-1.3815	0.3356	-0.1680	0.0627	84.5359	53.2354	-15.46%	6.47%	6.47%	6.39%	0.00%	0.00%	
24	3	1.1933	-0.8405	1.1933	0.1369	0.1538	0.0180	116.6310	50.9079	16.63%	1.82%	16.63%	16.42%	1.82%	1.79%	
25	4	0.2509	-0.3084	0.2509	-0.0713	0.0360	-0.0289	103.6703	48.5780	3.67%	-2.84%	3.67%	3.62%	0.00%	0.00%	
26	5	-0.4047	-0.3212	-0.4047	-0.5072	-0.0459	-0.1269	95.5139	44.0400	-4.49%	-11.92%	0.00%	0.00%	0.00%	0.00%	
50017	49996	-0.5600	-0.6834	-0.5600	-0.8833	-0.0653	-0.2116	93.6780	40.4660	-6.32%	-19.07%	0.00%	0.00%	0.00%	0.00%	
50018	49997	0.4625	-0.4862	0.4625	-0.0689	0.0625	-0.0283	106.4498	48.6046	6.45%	-2.79%	6.45%	6.37%	0.00%	0.00%	
50019	49998	-1.1020	-0.3230	-1.1020	-0.9617	-0.1331	-0.2292	87.5413	39.7586	-12.46%	-20.48%	0.00%	0.00%	0.00%	0.00%	
50020	49999	2.5944	0.4642	2.5944	2.0391	0.3290	0.4460	138.9558	78.1017	38.96%	56.20%	56.20%	55.51%	38.96%	38.47%	
50021	50000	-0.7516	-0.5744	-0.7516	-0.9250	-0.0893	-0.2209	91.4609	40.0882	-8.54%	-19.82%	0.00%	0.00%	0.00%	0.00%	
50022												Average	11.27%	Average	3.70%	

Once the terminal values have been calculated, we can use the continuous compounding discounting equation to determine each possible present value for the Type

4 option and each possible present value for the Type 6 option. We will then run many simulations and take an average. Notice that when the returns have a correlation of 0.65, Type 4 options are worth about 11.27% of the notional principal and Type 6 options are worth about 3.70% of the notional principal (these values will vary a bit each time the simulator is used). From your own simulation, you should see that changing the correlation between the stocks will change the results. Clearly, correlation must be factored into the valuation methodology, and Monte Carlo simulation makes this relatively easy to do.

As demonstrated by Kolb (2007), all other things being equal, the higher the correlation between two asset's returns, the lower the value of a Type 4 rainbow. For a Type 6 rainbow, the opposite is true.



This model can easily be adapted to handle any of the various types of rainbows we described at the beginning of this section, other types of rainbows, and rainbows of as many "colors" as we like.

Given the incredible flexibility of Monte Carlo simulation, it is no wonder that it has become a tool of choice for valuing complex derivatives.

Bibliography

Black, Fischer and Myron Scholes 1973: "The Pricing of Options and Corporate Liabilities." *Journal of Political Economy* 81:3, 637-654.

Briys, Eric, Mondher Bellalah, Huu Minh Mai, and Francois de Varenne 1998: *Options, Futures and Exotic Derivatives, Theory, Application and Practice*. New York: John Wiley & Sons.

Cox, John C., Stephen A. Ross, and Mark Rubinstein 1979: "Option Pricing: A Simplified Approach." *Journal of Financial Economics* 7, 229-263.

Hull, John 2005: *Options, Futures & Other Derivatives*, 6th edition. Upper Saddle River, NJ: Wiley.

Kolb, Robert W. 2007: *Futures, Options, & Swaps*, 5th edition. Malden, MA: Blackwell Publishing Limited.

McLeish, Don L. 2005: *Monte Carlo Simulation and Finance*. New York: John Wiley & Sons.

Smithson, Charles W. 1998: *Managing Financial Risk: A Guide to Derivative Products, Financial Engineering, and Value Maximization*, 3rd edition. New York: McGraw-Hill.